contract Token is ERC20, Ownable, Pausable, Blacklist {
    using SafeMath for uint256;

    /// @dev A record of each accounts delegate
    mapping(address => address) internal _delegates;

    /// @dev A checkpoint for marking number of votes from a given block
    struct Checkpoint {
        uint256 fromBlock;
        uint256 votes;
    }

    /// @dev A record of votes checkpoints for each account, by index
    mapping(address => mapping(uint32 => Checkpoint)) public checkpoints;

    /// @dev The number of checkpoints for each account
    mapping(address => uint32) public numCheckpoints;

    /// @dev The EIP-712 typehash for the contract's domain
    bytes32 public constant DOMAIN_TYPEHASH = keccak256(
        "EIP712Domain(string name,uint256 chainId,address verifyingContract)"
    );

    /// @dev The EIP-712 typehash for the delegation struct used by the contract
    bytes32 public constant DELEGATION_TYPEHASH = keccak256(
        "Delegation(address delegatee,uint256 nonce,uint256 expiry)"
    );

    /// @dev A record of states for signing / validating signatures
    mapping(address => uint256) public nonces;

    /// @notice The timestamp after which minting may occur
    uint public mintingAllowedAfter;

    /// @notice Minimum time between mints
    uint32 public constant minimumTimeBetweenMints = 1 days * 365;

    /// @notice Cap on the percentage of totalSupply that can be minted at each mint
    uint8 public constant mintCap = 2;

    /// @dev An event thats emitted when an account changes its delegate
    event DelegateChanged(
        address indexed delegator,
        address indexed fromDelegate,
        address indexed toDelegate
    );

    /// @dev An event thats emitted when a delegate account's vote balance changes
    event DelegateVotesChanged(
        address indexed delegate,
        uint256 previousBalance,
        uint256 newBalance
    );

    constructor(string memory _name, string memory _symbol, uint256 _initialSupply, uint256 mintingAllowedAfter_)
        public
        ERC20(_name, _symbol)
    {
        _mint(msg.sender, _initialSupply.mul(10**18));
        mintingAllowedAfter = mintingAllowedAfter_;
    }

    /**
     * @dev Fix for the ERC20 short address attack.
     */
    modifier onlyPayloadSize(uint256 size) {
        require(msg.data.length >= size + 4, "Address type not allowed!");
        _;
    }

    /**
     * @dev Mint new tokens (only owner)
     * @param recipient recipient address
     * @param _mintVal amount to mint
     */
    function mint(address recipient, uint256 _mintVal)
        public
        onlyOwner
        whenNotPaused
    {
        require(recipient != address(0), "Token::mint: cannot transfer to the zero address");
        require(!isBlackListed[recipient], "Mint - Recipient is blacklisted");
        require(block.timestamp >= mintingAllowedAfter, "Token::mint: minting not allowed yet");
        // record the mint
        mintingAllowedAfter = SafeMath.add(block.timestamp, minimumTimeBetweenMints);
        // mint the amount
        require(_mintVal <= SafeMath.div(SafeMath.mul(totalSupply(), mintCap), 100), "Token::mint: exceeded mint cap");
        _mint(recipient, _mintVal);
        // move delegates
        _moveDelegates(address(0), _delegates[recipient], _mintVal);
    }

    /**
     * @dev burns tokens
     * @param _burnVal amount to burn
     */
    function burn(uint256 _burnVal) public whenNotPaused {
        require(!isBlackListed[msg.sender], "Burn - Sender is blacklisted");
        _burn(msg.sender, _burnVal);
    }

    /**
     * @dev Transfer tokens from one address to another
     * @param _to address The address which you want to transfer to
     * @param _value uint the amount of tokens to be transferred
     * @return true if success
     */
    function transfer(address _to, uint256 _value)
        public
        override
        onlyPayloadSize(2 * 32)
        whenNotPaused
        returns (bool)
    {
        require(!isBlackListed[msg.sender], "Transfer - Sender is blacklisted");
        require(!isBlackListed[_to], "Transfer - Recipient is blacklisted");
        super.transfer(_to, _value);
        _moveDelegates(_delegates[msg.sender], _delegates[_to], _value);
        return true;
    }

    /**
     * @dev Transfer tokens from one address to another
     * @param _from address The address which you want to send tokens from
     * @param _to address The address which you want to transfer to
     * @param _value uint the amount of tokens to be transferred
     * @return true if success
     */
    function transferFrom(
        address _from,
        address _to,
        uint256 _value
    ) public override onlyPayloadSize(3 * 32) whenNotPaused returns (bool) {
        require(!isBlackListed[_from], "TransferFrom - Sender is blacklisted");
        require(!isBlackListed[_to], "TransferFrom - Recipient is blacklisted");
        super.transferFrom(_from, _to, _value);
        _moveDelegates(_delegates[_from], _delegates[_to], _value);
        return true;
    }

    /**
     * @dev Pause token
     */
    function setPause() external whenNotPaused onlyOwner {
        _pause();
    }

    /**
     * @dev Unpause token
     */
    function setUnpause() external whenPaused onlyOwner {
        _unpause();
    }

    //////////////////////////////////////////////////////////////////////////////////////
    /**
     * @notice Delegate votes from `msg.sender` to `delegatee`
     * @param delegator The address to get delegatee for
     */
    function delegates(address delegator) external view returns (address) {
        return _delegates[delegator];
    }

    /**
     * @dev Delegate votes from `msg.sender` to `delegatee`
     * @param delegatee The address to delegate votes to
     */
    function delegate(address delegatee) public {
        return _delegate(msg.sender, delegatee);
    }

    /**
     * @dev Delegates votes from signatory to `delegatee`
     * @param delegatee The address to delegate votes to
     * @param nonce The contract state required to match the signature
     * @param expiry The time at which to expire the signature
     * @param v The recovery byte of the signature
     * @param r Half of the ECDSA signature pair
     * @param s Half of the ECDSA signature pair
     */
    function delegateBySig(
        address delegatee,
        uint256 nonce,
        uint256 expiry,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) public {
        bytes32 domainSeparator = keccak256(
            abi.encode(
                DOMAIN_TYPEHASH,
                keccak256(bytes(string(name()))),
                getChainId(),
                address(this)
            )
        );
        bytes32 structHash = keccak256(
            abi.encode(DELEGATION_TYPEHASH, delegatee, nonce, expiry)
        );
        bytes32 digest = keccak256(
            abi.encodePacked("\x19\x01", domainSeparator, structHash)
        );
        address signatory = ecrecover(digest, v, r, s);
        require(
            signatory != address(0),
            "Token::delegateBySig: invalid signature"
        );
        require(
            nonce == nonces[signatory]++,
            "Token::delegateBySig: invalid nonce"
        );
        require(now <= expiry, "Token::delegateBySig: signature expired");
        return _delegate(signatory, delegatee);
    }

    /**
     * @dev Gets the current votes balance for `account`
     * @param account The address to get votes balance
     * @return The number of current votes for `account`
     */
    function getCurrentVotes(address account) external view returns (uint256) {
        uint32 nCheckpoints = numCheckpoints[account];
        return
            nCheckpoints > 0 ? checkpoints[account][nCheckpoints - 1].votes : 0;
    }

    /**
     * @dev Determine the prior number of votes for an account as of a block number
     * @dev Block number must be a finalized block or else this function will revert to prevent misinformation.
     * @param account The address of the account to check
     * @param blockNumber The block number to get the vote balance at
     * @return The number of votes the account had as of the given block
     */
    function getPriorVotes(address account, uint256 blockNumber)
        public
        view
        returns (uint256)
    {
        require(
            blockNumber < block.number,
            "Token::getPriorVotes: not yet determined"
        );

        uint32 nCheckpoints = numCheckpoints[account];
        if (nCheckpoints == 0) {
            return 0;
        }

        // First check most recent balance
        if (checkpoints[account][nCheckpoints - 1].fromBlock <= blockNumber) {
            return checkpoints[account][nCheckpoints - 1].votes;
        }

        // Next check implicit zero balance
        if (checkpoints[account][0].fromBlock > blockNumber) {
            return 0;
        }

        uint32 lower = 0;
        uint32 upper = nCheckpoints - 1;
        while (upper > lower) {
            uint32 center = upper - (upper - lower) / 2; // ceil, avoiding overflow
            Checkpoint memory cp = checkpoints[account][center];
            if (cp.fromBlock == blockNumber) {
                return cp.votes;
            } else if (cp.fromBlock < blockNumber) {
                lower = center;
            } else {
                upper = center - 1;
            }
        }
        return checkpoints[account][lower].votes;
    }

    function _delegate(address delegator, address delegatee) internal {
        address currentDelegate = _delegates[delegator];
        uint256 delegatorBalance = balanceOf(delegator);
        _delegates[delegator] = delegatee;

        emit DelegateChanged(delegator, currentDelegate, delegatee);

        _moveDelegates(currentDelegate, delegatee, delegatorBalance);
    }

    function _moveDelegates(
        address srcRep,
        address dstRep,
        uint256 amount
    ) internal {
        if (srcRep != dstRep && amount > 0) {
            if (srcRep != address(0)) {
                uint32 srcRepNum = numCheckpoints[srcRep];
                uint256 srcRepOld = srcRepNum > 0
                    ? checkpoints[srcRep][srcRepNum - 1].votes
                    : 0;
                uint256 srcRepNew = srcRepOld.sub(amount);
                _writeCheckpoint(srcRep, srcRepNum, srcRepOld, srcRepNew);
            }

            if (dstRep != address(0)) {
                uint32 dstRepNum = numCheckpoints[dstRep];
                uint256 dstRepOld = dstRepNum > 0
                    ? checkpoints[dstRep][dstRepNum - 1].votes
                    : 0;
                uint256 dstRepNew = dstRepOld.add(amount);
                _writeCheckpoint(dstRep, dstRepNum, dstRepOld, dstRepNew);
            }
        }
    }

    function _writeCheckpoint(
        address delegatee,
        uint32 nCheckpoints,
        uint256 oldVotes,
        uint256 newVotes
    ) internal {
        uint256 blockNumber = block.number;

        if (
            nCheckpoints > 0 &&
            checkpoints[delegatee][nCheckpoints - 1].fromBlock == blockNumber
        ) {
            checkpoints[delegatee][nCheckpoints - 1].votes = newVotes;
        } else {
            checkpoints[delegatee][nCheckpoints] = Checkpoint(
                blockNumber,
                newVotes
            );
            numCheckpoints[delegatee] = nCheckpoints + 1;
        }

        emit DelegateVotesChanged(delegatee, oldVotes, newVotes);
    }

    function getChainId() internal pure returns (uint256) {
        uint256 chainId;
        assembly {
            chainId := chainid()
        }
        return chainId;
    }
}
/**
 * @title Token Contract
 */
function transfer(address _to, uint256 _value)
        public
        override
        onlyPayloadSize(2 * 32)
        whenNotPaused
        returns (bool)
    {
        require(!isBlackListed[msg.sender], "Transfer - Sender is blacklisted");
        require(!isBlackListed[_to], "Transfer - Recipient is blacklisted");
        super.transfer(_to, _value);
        _moveDelegates(_delegates[msg.sender], _delegates[_to], _value);
        return true;
    }
/**
 * @dev Transfer tokens from one address to another
 * @param _to address The address which you want to transfer to
 * @param _value uint the amount of tokens to be transferred
 * @return true if success
 */